Kalman example

Let k = K[t, R1, R2, L, C] be the commutative polynomial ring in t, R1, R2, L, C and let us define the Ore algebra A = k⟨∂⟩, where ∂ is differential operator defined by $\partial f(t) = f'(t)$.

```
A = OreAlgebra[t, Der[t], R1, R2, L, C]
```

```
𝕂[t, R1, R2, L, C][Dt; 1, Dt]
```

We consider the matrix $R \in A^{2\times 3}$ given by

```
MatrixForm[R = ToOrePolynomial[{{R1 C Der[t] + 1, 0, -1},
     {0, L Der[t] + R2, -1}}, A]]
```

$$\begin{pmatrix} D_t\,R1\,C + 1 & 0 & -1 \\ 0 & D_t\,L + R2 & -1 \end{pmatrix}$$

the system defined by

```
Thread[ApplyMatrix[R, {x1[t], x2[t], u[t]}] == 0]
```

$$\{-u[t] + x_1[t] + C\,R1\,x_1'[t] == 0,\ -u[t] + R2\,x_2[t] + L\,x_2'[t] == 0\}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 3} / A^{1\times 2} R$.

Let us compute the adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t\,R1\,C + 1 & 0 \\ 0 & -D_t\,L + R2 \\ -1 & -1 \end{pmatrix}$$

Let us check whether or not the left A-module M is torsion-free, i.e. whether or not the system admits some autonomous elements.

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & D_t\,L + R2 & -1 \\ -D_t\,R1\,C - 1 & 0 & 1 \end{pmatrix}, \begin{pmatrix} D_t\,L + R2 \\ D_t\,R1\,C + 1 \\ D_t^2\,R1\,L\,C + D_t\,R1\,R2\,C + D_t\,L + R2 \end{pmatrix}\right\}$$

Since the first matrix is identity, M is torsion-free. Let us now compute the obstructions for M to be projective:

```
P = ObstructionToProjectiveness[R, A]
```

$$\{\{\{D_t\,L + R2\},\ \{R1\,R2\,C - L\},\ \{D_t\,R1\,C + 1\}\}\}$$

Let us deduce the obstructions in the parameters R1, R2, L, C for M to be a projective left A-module.

```
OreIntersection[Flatten@P, {R1, R2,  L,  C}, A]
```

$$\{R1\,R2\,C - L\}$$

We obtain that M is a projective left A-module, i.e. the system is flat, if and only if $C\,R1\,R2 - L$ is not equal to zero.

If C R1 R2-L ≠ 0, let us compute flat output.

```
B = OreAlgebra[t, Der[t]]
```

```
𝕂[t][Dt; 1, Dt]
```

```
T = LeftInverse[Q, B]
```

$$\left\{\left\{\frac{C\,R1}{-L+C\,R1\,R2}, \frac{L}{L-C\,R1\,R2}, 0\right\}\right\}$$

A flat output ξ is then defined by:

```
ApplyMatrix[T, {x1[t], x2[t], u[t]}][[1]]
```

$$\frac{-C\,R1\,x_1[t]+L\,x_2[t]}{L-C\,R1\,R2}$$

Finally a parametrization of the flat system is defined by

```
Thread[{x1[t], x2[t], u[t]} -> ApplyMatrix[Q, {ξ[t]}]]
```

$$\{x_1[t]\to R2\,\xi[t]+L\,\xi'[t],\ x_2[t]\to \xi[t]+C\,R1\,\xi'[t],\ u[t]\to R2\,\xi[t]+(L+C\,R1\,R2)\,\xi'[t]+C\,L\,R1\,\xi''[t]\}$$

If C R1 R2-L = 0, then let us compute the autonomous elements of the new system. To do that let us introduce a new Ore algebra B, defined by

```
B = OreAlgebra[t, Der[t], CoefficientNormal → (Expand[# /. L → C R1 R2] &)]
```

$\mathbb{K}[t][D_t;\ 1,\ D_t]$

the matrix S of differential operators given by

```
MatrixForm[S = ToOrePolynomial[{{R1 C Der[t] + 1, 0, -1},
    {0, L Der[t] + R2, -1}}, B]]
```

$$\begin{pmatrix} C\,R1\,D_t+1 & 0 & -1 \\ 0 & C\,R1\,R2\,D_t+R2 & -1 \end{pmatrix}$$

and the left B-module N finitely presented by S. Let us compute the torsion elements of N, i.e. the autnomous elements of the corresponding system.

```
MatrixForm[Sadj = Involution[S, B]]
```

$$\begin{pmatrix} -C\,R1\,D_t+1 & 0 \\ 0 & -C\,R1\,R2\,D_t+R2 \\ -1 & -1 \end{pmatrix}$$

```
MatrixForm /@ ({Ann, Sp, Q} = Exti[Sadj, B, 1])
```

$$\left\{\begin{pmatrix} -C\,R1\,D_t-1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} -1 & R2 & 0 \\ 0 & C\,R1\,R2\,D_t+R2 & -1 \end{pmatrix}, \begin{pmatrix} R2 \\ 1 \\ C\,R1\,R2\,D_t+R2 \end{pmatrix}\right\}$$

Since the first matrix is not identity, the system is not controllable (N is not torsion free) and we have one autonomous element, defined by the first row of Sp, namely θ, defined by

```
ApplyMatrix[{Sp[[1]]}, {x1[t], x2[t], u[t]}]
```

$$\{-x_1[t]+R2\,x_2[t]\}$$

Moreover, θ satisfies the equation

```
Thread[ApplyMatrix[{{Ann[[1, 1]]}}, {θ[t]}] == 0]
```

$$\{-\theta[t]-C\,R1\,\theta'[t]==0\}$$

Finally, this result can be obtained directly by means of the command AutonomousElements

```
AutonomousElements[S, {x1[t], x2[t], u[t]}, θ, B]
```

```
{{θ[1][t] → -x1[t] + R2 x2[t], θ[2][t] → -u[t] + R2 (x2[t] + C R1 x2'[t])},
 {-θ[1][t] - C R1 θ[1]'[t] == 0, θ[2][t] == 0}}
```

The controllable part of the system is defined by:

```
Thread[ApplyMatrix[Sp, {x1[t], x2[t], u[t]}] == 0]
```

```
{-x1[t] + R2 x2[t] == 0, -u[t] + R2 (x2[t] + C R1 x2'[t]) == 0}
```

Finally let us compute a flat output of the controllable part:

```
ApplyMatrix[LeftInverse[Q, B], {x1[t], x2[t], u[t]}][[1]]
```

```
x2[t]
```